
HP-35s Calculator Program     Utility 1 

HMS+ 

Programmer: Dr. Bill Hazelton 

Date: October, 2007. 
 

Line Instruction Display User Instructions 
C001 LBL C  Enter first angle. Press  ENTER. 
C002 HMS→  Enter second angle. Press  XEQ  C  ENTER. 
C003 x < > y   
C004 HMS→  (Angles must be in DDD.MMSS format) 
C005 +   
C006 →HMS   
C007 RTN  Angle sum displayed (in HP notation) 

Notes 

 (1) General program to add two angles, azimuths or directions in DDD.MMSS 
format (HP notation), and produce a result in the same format. 

 (2) Key in the first angle. Press ENTER. Key in the second angle. Stack will contain: 
 

Stack Register Contents 

T  

Z  

Y First angle in D.MS (line 1) 

X Second angle in D.MS (line 2) 

  Press XEQ  C  ENTER. The sum of the two angles in HP notation will be in the 
X register (line 1) 

 (3) Negative values will work correctly. 

Sample Computation 

123° 45' 56"  +  321° 54' 32"  =  445° 40' 28" 

Storage Registers Used 

 None 

Labels Used 

Label C  Length  =   21  Checksum  =  F341 

Use the length (LN=) and Checksum (CK=) values to check if program was entered correctly. 
Use the sample computation to check proper operation after entry. 



© 2007  N.W.J. Hazelton 

HP-35s Calculator Program     Utility 2 

HMS– 

Programmer: Dr. Bill Hazelton 

Date: October, 2007. 
 

Line Instruction Display User Instructions 
D001 LBL D  Enter first angle. Press  ENTER. 
D002 HMS→  Enter second angle. Press  XEQ  D  ENTER. 
D003 x < > y   
D004 HMS→  (Angles in DDD.MMSS format) 
D005 x < > y   
D006 –   
D007 →HMS   
D008 RTN  Angle sum displayed (in HP notation) 

Notes 
 (1) General program to get the difference between two angles, azimuths or directions 

in DDD.MMSS format (HP notation), and produce a result in the same format. 

 (2) Key in the first angle. Press ENTER. Key in the second angle. Stack will contain: 
 

Stack Register Contents 

T  

Z  

Y First angle in DMS (line 1) 

X Second angle in DMS (line 2) 

  Press XEQ  D  ENTER. The difference between the two angles in HP notation 
will be in the X register. The second angle will be subtracted from the first. 

 (3) Negative values will work correctly. 

Sample Computation 

321° 54' 32"  –  123° 45' 56"    =  198° 08' 36" 

Storage Registers Used 

 None 

Labels Used 

Label D  Length  =   24  Checksum  =  E0E0 

Use the length (LN=) and Checksum (CK=) values to check if program was entered correctly. 
Use the sample computation to check proper operation after entry. 

HP-35s Calculator Program     Utility 3 

Enter Vector with D.MMSS Azimuth for Complex Number 

Programmer: Dr. Bill Hazelton 

Date: October, 2007.   Mnemonic: V for Vector Building 
 

Line Instruction Display User Instructions 
V001 LBL V  Press  XEQ  V  ENTER. 
V002 10   
V003 STO  I   
V004 RCL  A   
V005 STO  (I)    STO  (I)   ( (I) is on the zero key) 
V006 1   
V007 STO +  I   
V008 RCL  D   
V009 STO  (I)   
V010 1   
V011 STO +  I   
V012 INPUT  A A? Prompts for azimuth in D.MMSS 
V013 RCL  A   
V014 HMS→   
V015 STO  A   
V016 INPUT  D D? Prompts for distance (D?) 
V017 RCL  A   
V018 COS   
V019 RCL ×  D   
V020 RCL  A   
V021 SIN   
V022 RCL ×  D   
V023 0 i 1  [Key in as 0, then i, then 1, press ENTER] 
V024 ×   
V025 +   
V026 STO  (I)   
V027 10   
V028 STO  I   
V029 RCL  (I)   
V030 STO  A   
V031 1   
V032 STO +  I   
V033 RCL (I)   
V034 STO  D   
V035 1   
V036 STO +  I   
V037 CLSTK   
V038 RCL (I)   
V039 RTN  Complex number now in stack in X 

Notes 

 (1) A program that allows the user to enter a vector as two separate components, 
azimuth and distance, with the azimuth in D.MMSS (HP format), and have it 
converted to a complex number, with the azimuth component in decimal degrees. 

 (2) Calculator should be set in DEGREES mode. Press MODE, then 1. 

 (3) Because the calculator uses data entry into the A and D storage registers to allow 
simple prompting, there is the potential of this program deleting any data already 
in those storage registers. This could be a problem if this program was called as a 
sub-routine from within another program. To avoid this problem, the program 
copies the contents of storage register A to storage register 10, and the contents 
of storage register D to storage register 11. At the conclusion of the program, the 
values are copied back into storage registers A and D. 

 (4) Because the program replaces everything that was on the stack before it ran, the 
stack is cleared of all data before the result of the calculation is returned to the 
stack. 

 (5) While copying back the contents of registers 10 and 11, the program stores the 
result of the calculation in storage register 12. 

 (6) If using a program that requires storage registers 10, 11 and 12, change the value 
of 10 in lines V002 and V027 to a suitable number, so that the set of three storage 
registers selected aren’t used elsewhere. 

Operation 

 Press  XEQ  V  ENTER. 

The calculator prompts with A? to enter the azimuth in degrees, minutes and seconds (HP 
notation). Key in the azimuth and press R/S. 

The calculator prompts with D? to enter the distance. Key in the distance and press R/S. 

The calculator displays the complex number representing the vector in the X register, or 
line 2 of the display. The remainder of the stack is zeros. 

Examples 

1. 63° 15' 47"  and 105.528 will give a complex number of  105.5280 θ 63.2631 in 
polar mode (with FIX 4 set for the display), or  47.4765 i 94.2451 in rectangular 
mode. 

2. 128° 15' 47"  and 105.528 will give a complex number of  105.5280 θ 128.2631, 
or –65.3506 i 82.8580. 

3. 237° 15' 47"  and 105.528 will give a complex number of 105.5280 θ –122.7369, 
or  –57.0677 i –88.7662. 

4. 333° 15' 47"  and 105.528 will give a complex number of 105.5280 θ –26.7369, 
or  94.2451 i –47.4765. 

5. 397° 15' 47" and 105.528 produce 105.5280 θ 37.2631, or 83.9859 i 63.8946. 

Storage Registers Used 

 A Stores the azimuth value, initially in D.MMSS, then in decimal degrees. 

 D Stores the distance value. 

 I Used to store the value for indirect addressing of registers 10, 11 and 12. 

 10 Temporary storage for the contents of storage register A while the program runs. 

 11 Temporary storage for the contents of storage register D while the program runs. 

 12 Temporary storage for the answer while copying back A and D. 

Labels Used 

Label V  Length  =   128  Checksum  =  39FE 

Use the length (LN=) and Checksum (CK=) values to check if program was entered correctly. 
Use the sample computation to check proper operation after entry. 

HP-35s Calculator Program     Utility 4 

Extract Real and Imaginary Parts of a Complex Number 

Programmer: Dr. Bill Hazelton 

Date: October, 2007.   Mnemonic: X for eXtract Co-ordinates 
 

Line Instruction Display User Instructions 
X001 LBL X  Press  XEQ  X  ENTER. 
X002 10   
X003 STO  I   
X004 R↓   
X005 STO  (I)    STO  (I)   ( (I) is on the zero key) 
X006 ARG   
X007 COS   
X008 RCL  (I)   
X009 ABS   
X010 ×   
X011 RCL  (I)   
X012 ARG   
X013 SIN   
X014 RCL  (I)   
X015 ABS   
X016 ×   
X017 RTN  Components now on stack 

Notes 

 (1) The program is designed to operate with the complex number on the stack in the 
X register (line 2 of the display). 

 (2) The program stores the vector in storage register 10 for quick retrieval during 
operation. If this storage register is in use, change the value in line X002 to a 
suitable number that doesn’t clash with other needs. 

(3) The program will work regardless of the display mode for complex numbers, and 
regardless of the angle unit mode of the calculator. 

(4) The program will over-write values in the stack prior to it being called. 

Operation 

(1) Put the complex number into the X register on the stack, either by keying it in, or 
by recalling it there from wherever it is stored. 

(2) Press  XEQ  X  ENTER. The program returns the real (or Northing or Y) 
component or co-ordinate to the Y register (line 1 of the display), and the 
imaginary part or co-ordinate (or Easting or X) to the X register (line 2 of the 
display). 

Examples 

1. The complex number 123.0000 i 456.0000 is on the stack. The result is 123.0000 
in the Y register and 456.000 in the X register. 

2. The complex number 100.0000 θ 45.0000 is on the stack. The result is 70.7107 
in the Y register and 70.7107 in the X register. 

3. The complex number 100.0000 θ 230.0000 is on the stack. The result is –64.2788 
in the Y register and –76.6044 in the X register. 

Storage Registers Used 

 I Used for the indirect addressing of storage register 10. 

 10 Used to store the complex number during operations. 

Labels Used 

Label X  Length  =   53  Checksum  =  C46D 

Use the length (LN=) and Checksum (CK=) values to check if program was entered correctly. 
Use the sample computation to check proper operation after entry. 

HP-35s Calculator Program     Utility 5 

Enter Vector with D.MMSS Azimuth for Complex Number, 
but check for a negative distance 

Programmer: Dr. Bill Hazelton 

Date: December, 2007.   Mnemonic: W for Vector Wrangling 
 

Line Instruction Display User Instructions 
W001 LBL W  Press  XEQ  W  ENTER. 
W002 10   
W003 STO  I   
W004 RCL  A   
W005 STO  (I)    STO  (I)   ( (I) is on the zero key) 
W006 1   
W007 STO +  I   
W008 RCL  D   
W009 STO  (I)   
W010 1   
W011 STO +  I   
W012 INPUT  A A? Prompts for azimuth in D.MMSS 
W013 RCL  A   
W014 HMS→   
W015 STO  A   
W016 INPUT  D D? Prompts for distance (D?) 
W017 RCL  D   
W018 x ≥ 0?   
W019 GTO  W023   
W020 ABS   
W021 STO  D   
W022 SF  3   
W023 RCL  A   
W024 COS   
W025 RCL ×  D   
W026 RCL  A   
W027 SIN   
W028 RCL ×  D   
W029 0 i 1  [Key in as 0, then i, then 1, press ENTER] 
W030 ×   
W031 +   
W032 STO  (I)   
W033 10   
W034 STO  I   
W035 RCL  (I)   
W036 STO  A   
W037 1   
W038 STO +  I   

W039 RCL (I)   
W040 STO  D   
W041 1   
W042 STO +  I   
W043 CLSTK   
W044 RCL (I)   
W045 RTN  Complex number now in stack in X 

Special Note 

This program is designed to work with the B program (Closure 5) that computes a traverse 
closure and area, but allows for curves in the polygon boundary. In the Closure 5 program, the 
way to signal that an entered side is a chord is to enter a negative distance. This would ordinarily 
result in a complex number that was perfectly valid for computation, which meant that there was 
no simple way to signal the entry of a chord. The V program (Utility 3) was modified to check if 
the distance was negative, and if so, to set Flag 3, take the absolute value of the distance, and then 
proceed as before. The Closure 5 program checks if Flag 3 is set when the W sub-program 
returns, and processes the line based on the state of Flag 3. 

Notes 

 (1) A program that allows the user to enter a vector as two separate components, 
azimuth and distance, with the azimuth in D.MMSS (HP format), and have it 
converted to a complex number, with the azimuth component in decimal degrees. 
If the distance entered is negative, the value is made positive and Flag 3 is set to 
indicate the negative distance. 

 (2) Calculator should be set in DEGREES mode. Press MODE, then 1. 

 (3) Because the calculator uses data entry into the A and D storage registers to allow 
simple prompting, there is the potential of this program deleting any data already 
in those storage registers. This could be a problem if this program was called as a 
sub-routine from within another program. To avoid this problem, the program 
copies the contents of storage register A to storage register 10, and the contents 
of storage register D to storage register 11. At the conclusion of the program, the 
values are copied back into storage registers A and D. 

 (4) Because the program replaces everything that was on the stack before it ran, the 
stack is cleared of all data before the result of the calculation is returned to the 
stack. 

 (5) While copying back the contents of registers 10 and 11, the program stores the 
result of the calculation in storage register 12. 

 (6) If using a program that requires storage registers 10, 11 and 12, change the value 
of 10 in lines W002 and W033 to a suitable number, so that the set of three 
storage registers selected aren’t used elsewhere. 

Operation 
 Press  XEQ  W  ENTER. 

The calculator prompts with A? to enter the azimuth in degrees, minutes and seconds (HP 
notation). Key in the azimuth and press R/S. 

The calculator prompts with D? to enter the distance. Key in the distance and press R/S. 

The calculator displays the complex number representing the vector in the X register, or 
line 2 of the display. The remainder of the stack is zeros. 

Examples 
1. 63° 15' 47"  and 105.528 will give a complex number of  105.5280 θ 63.2631 in 

polar mode (with FIX 4 set for the display), or  47.4765 i 94.2451 in rectangular 
mode. 

2. 128° 15' 47"  and 105.528 will give a complex number of  105.5280 θ 128.2631, 
or –65.3506 i 82.8580. 

3. 237° 15' 47"  and 105.528 will give a complex number of 105.5280 θ –122.7369, 
or  –57.0677 i –88.7662. 

4. 333° 15' 47"  and 105.528 will give a complex number of 105.5280 θ –26.7369, 
or  94.2451 i –47.4765. 

5. 397° 15' 47" and 105.528 will produce 105.5280 θ 37.2631, or 83.9859 i 
63.8946. 

6. 63° 15' 47"  and –105.528 will give a complex number of  105.5280 θ 63.2631 in 
polar mode, or  47.4765 i 94.2451 in rectangular mode. This is not what would 
be expected in normal mathematical work, but a consequence of always taking 
the distance to be positive. 

7. 128° 15' 47"  and –105.528 will give a complex number of  105.5280 θ 
128.2631, or –63.3506 i 82.8580. Again, not standard mathematically. 

Storage Registers Used 

 A Stores the azimuth value, initially in D.MMSS, then in decimal degrees. 

 D Stores the distance value. 

 I Used to store the value for indirect addressing of registers 10, 11 and 12. 

 10 Temporary storage for the contents of storage register A while the program runs. 

 11 Temporary storage for the contents of storage register D while the program runs. 

 12 Temporary storage for the answer while copying back A and D. 

Labels Used 

Label W  Length  =   146  Checksum  =  A1A3 

Use the length (LN=) and Checksum (CK=) values to check if program was entered correctly. 
Use the sample computation to check proper operation after entry. 


